// File: lib/openzeppelin-contracts/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: lib/openzeppelin-contracts/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: lib/openzeppelin-contracts/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: src/Deadjira.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.17;

import {Ownable} from "open-zeppelin/contracts/access/Ownable.sol";
import {Pausable} from "open-zeppelin/contracts/security/Pausable.sol";

/// @title DeadjiraAuction
/// @author WhiteOakKong
/// @notice A contract for conducting an auction of limited edition Deadjira ordinals.
/// @dev Inherits from the OpenZeppelin Ownable and Pausable contracts.

contract DeadjiraAuction is Ownable, Pausable {
    uint256 public constant AUCTION_SUPPLY = 50;
    uint256 public constant STARTING_PRICE = 3.33 ether;
    uint256 public constant ENDING_PRICE = 1.11 ether;
    uint256 private constant INCREMENT_VALUE = 0.11 ether;
    uint256 private constant INCREMENT_TIME = 213 seconds;

    uint256 public startTime;

    address public withdrawAddress;

    address[] public buyers;
    mapping(address => uint256) public purchased;

    /// @notice Event emitted when a new auction purchase is made.
    /// @param minter The address of the buyer.
    /// @param btcAddress The buyer's BTC address.
    /// @param discordID The buyer's Discord ID.
    /// @param purchaseID The unique ID of the purchase.
    event AuctionPurchase(address minter, string btcAddress, string discordID, uint256 purchaseID);

    /// @dev Constructor to initialize the contract and set it to a paused state.
    constructor() {
        _pause();
    }

    // USER FUNCTIONS //

    /// @notice Allows users to purchase a token during the auction.
    /// @dev Function can only be called when the contract is not paused.
    /// @param btcAddress The buyer's BTC address.
    /// @param discordID The buyer's Discord ID.
    function purchaseAuction(string calldata btcAddress, string calldata discordID) external payable whenNotPaused {
        require(msg.value >= calculatePrice(), "Incorrect value.");
        require(buyers.length < AUCTION_SUPPLY, "Auction supply reached.");
        require(purchased[msg.sender] == 0, "Already purchased.");
        buyers.push(msg.sender);
        purchased[msg.sender] = msg.value;
        emit AuctionPurchase(msg.sender, btcAddress, discordID, buyers.length);
    }

    // ACCESS CONTROLLED FUNCTIONS //

    /// @notice Starts the auction by setting the start time and unpausing the contract.
    /// @dev Can only be called by the contract owner.
    function startAuction() external onlyOwner {
        require(startTime == 0, "Auction already started.");
        _unpause();
        startTime = block.timestamp;
    }

    /// @notice Toggles the pause state of the contract.
    /// @dev Can only be called by the contract owner.
    function togglePause() external onlyOwner {
        require(startTime > 0, "Auction not started.");
        if (paused()) _unpause();
        else _pause();
    }

    /// @notice Allows the contract owner to withdraw the contract balance.
    /// @dev Can only be called when the contract is paused and by the contract owner.
    function withdraw() external whenPaused onlyOwner {
        require(withdrawAddress != address(0), "Withdraw address not set.");
        (bool success, ) = payable(withdrawAddress).call{value: address(this).balance}("");
        require(success, "Transfer failed.");
    }

    ///@notice Change the address to which funds can be withdrawn.
    ///@dev Can only be called by the contract owner and cannot be the zero address.
    ///@param _address The new address to set as the withdrawal address.
    function setWithdrawAddress(address _address) external onlyOwner {
        require(_address != address(0), "Invalid address.");
        withdrawAddress = _address;
    }

    /// @notice Allows the contract owner to issue refunds to buyers after the auction has ended.
    /// @dev Can only be called when the contract is paused and by the contract owner.
    /// @param finalPrice The final price of the tokens at the end of the auction.
    function refund(uint256 finalPrice) external whenPaused onlyOwner {
        require(finalPrice >= ENDING_PRICE, "Invalid final price.");
        address[] memory _buyers = buyers;
        for (uint256 i = 0; i < _buyers.length; i++) {
            uint256 refundValue = purchased[_buyers[i]] - finalPrice;
            if (refundValue > 0) {
                (bool success, ) = payable(_buyers[i]).call{value: refundValue}("");
                require(success, "Transfer failed.");
            }
        }
    }

    // VIEW FUNCTIONS //

    /// @notice Calculates the current price of the tokens based on the elapsed time since the auction started.
    /// @return The current price of the tokens.
    function calculatePrice() public view returns (uint256) {
        uint256 timeSinceStart = block.timestamp - startTime;
        uint256 increments = timeSinceStart / INCREMENT_TIME;
        if (increments * INCREMENT_VALUE > STARTING_PRICE - ENDING_PRICE) return ENDING_PRICE;
        uint256 price = STARTING_PRICE - (increments * INCREMENT_VALUE);
        return price;
    }

    /// @notice Retrieves the purchase data for a specific user.
    /// @param buyer The address of the buyer.
    /// @return The purchase value of the specified buyer.
    function getUserPurchaseData(address buyer) external view returns (uint256) {
        return purchased[buyer];
    }

    /// @notice Retrieves the total number of purchases made during the auction.
    /// @return The total number of purchases.
    function getTotalPurchased() external view returns (uint256) {
        return buyers.length;
    }

    receive() external payable {}
}